
5. Creating a Stand-Alone Visual Basic Application





Review and Preview



We've finished looking at most of the Visual Basic tools and been introduced to most of the Basic language features.  Thus far, to run any of the applications studied, we needed Visual Basic.  In this class, we learn the steps of developing a stand-alone application that can be run on any Windows-based machine.  We’ll also look at some new components that help make up applications.





Designing an Application



Before beginning the actual process of building your application by drawing the Visual Basic interface, setting the object properties, and attaching the Basic code, many things should be considered to make your application useful.



A first consideration should be to determine what processes and functions you want your application to perform.  What are the inputs and outputs?  Develop a framework or flow chart of all your application's processes. 



Decide what tools you need.  Do the built-in Visual Basic tools and functions meet your needs?  Do you need to develop some tools or functions of your own?



Design your user interface.  What do you want your form to look like?  Consider appearance and ease of use.  Make the interface consistent with other Windows applications.  Familarity is good in program design.



Write your code.  Make your code readable and traceable - future code modifiers will thank you.  Consider developing reusable code - modules with utility outside your current development.  This will save you time in future developments.



�Make your code 'user-friendly.'  Try to anticipate all possible ways a user can mess up in using your application.  It's fairly easy to write an application that works properly when the user does everything correctly.  It's difficult to write an application that can handle all the possible wrong things a user can do and still not bomb out.



Debug your code completely before distributing it.  There's nothing worse than having a user call you to point out flaws in your application.  A good way to find all the bugs is to let several people try the code - a mini beta-testing program.





Using General Sub Procedures in Applications



So far in this class, the only procedures we have studied are the event-driven procedures associated with the various tools.  Most applications have tasks not related to objects that require some code to perform these tasks.  Such tasks are usually coded in a general Sub procedure (essentially the same as a subroutine in other languages).  



Using general Sub procedures can help divide a complex application into more manageable units of code.  This helps meet the above stated goals of readability and reusability.



Defining a Sub Procedure:



The form for a general Sub procedure named GenlSubProc is:



Sub GenlSubProc(Arguments)	 'Definition header

	.

	.

End Sub



The definition header names the Sub procedure and defines any arguments passed to the procedure.  Arguments are a comma-delimited list of variables passed to and/or from the procedure. If there are arguments, they must be declared and typed in the definition header in this form:



Var1 As Type1, Var2 As Type2, ...



�Sub Procedure Example:



Here is a Sub procedure (USMexConvert) that accepts as inputs an amount in US dollars (USDollars) and an exchange rate (UStoPeso).  It then outputs an amount in Mexican pesos (MexPesos).



Sub USMexConvert (USDollars As Single, UStoPeso As Single, MexPesos As Single)

MexPesos = UsDollars * UsToPeso

End Sub



Calling a Sub Procedure:



There are two ways to call or invoke a Sub procedure.  You can also use these to call event-driven procedures.



Method 1:



Call GenlSubProc(Arguments) 	(if there are no Arguments, do not type the parentheses)



Method 2:



GenlSubProc Arguments



I prefer Method 1 - it's more consistent with calling protocols in other languages and it cleanly delineates the argument list.  It seems most Visual Basic programmers use Method 2, though.  I guess they hate typing parentheses!  Choose the method you feel more comfortable with.



Example



To call our dollar exchange routine, we could use:



Call USMexConvert (USDollars, UStoMex, MexPesos)



or



USMexConvert USDollars, UStoMex, MexPesos



�Locating General Sub Procedures:



General Sub procedures can be located in one of two places in your application:  attached to a form or attached to a module.  Place the procedure in the form if it has a purpose specifically related to the form.  Place it in a module if it is a general purpose procedure that might be used by another form or module or another application.



Whether placing the procedure in a form or module, the methods of creating the procedure are the same.  Select or open the form or module's code window.  Make sure the window's Object list says (General) and the Procedure list says (Declarations).  You can now create the procedure by selecting Add Procedure from Visual Basic's Tools menu.  A window appears allowing you to select Type Sub and enter a name for your procedure. Another way to create a Sub is to go to the last line of the General Declarations section, type Sub followed by a space and the name of your procedure.  Then, hit Enter.  With either method for establishing a Sub, Visual Basic will form a template for your procedure.  Fill in the Argument list and write your Basic code.  In selecting the Insert Procedure menu item, note another option for your procedure is Scope.  You have the choice of Public or Private.  The scope word appears before the Sub word in the definition heading.  If a module procedure is Public, it can be called from any other procedure in any other module.  If a module procedure is Private, it can only be called from the module it is defined in.  Note, scope only applies to procedures in modules.  By default, all event procedures and general procedures in a form are Private - they can only be called from within the form.  You must decide the scope of your procedures.



Passing Arguments to Sub Procedures:



A quick word on passing arguments to procedures.  By default, they are passed by reference.  This means if an argument is changed within the procedure, it will remain changed once the procedure is exited.



C programmers have experienced the concept of passing by value, where a parameter changed in a procedure will retain the value it had prior to calling the routine.  Visual Basic also allows calling by value.  To do this, place the word ByVal in front of each such variable in the Argument list.





�

Creating a Code Module



If you're going to put code in a module, you'll need to know how to create and save a module.  A good way to think about modules is to consider them forms without any objects, just code.



To create a module, click on the New Module button on the toolbar, or select the Module option from the Insert menu.  The module will appear.  Note any modules appear in the Project Window, along with your form(s).  You use the Project Window to move among forms and modules.



Once the module is active, establish all of your procedures as outlined above.  To name the module, click on the properties window while the module is active.  Note Name is the only property associated with a module.  Saving a module is just like saving a form - use the Save File and Save File As options.







Using General Function Procedures in Applications



Related to Sub procedures are Function procedures.  A Function procedure, or simply Function, performs a specific task within a Visual Basic program and returns a value.  We've seen some built-in functions such as the MsgBox and the Format function.



Defining a Function:



The form for a general Function named GenlFcn is:



Function GenlFcn(Arguments) As Type  'Definition header

	.

	.

GenlFcn = ...

End Function



The definition header names the Function and specifies its Type (the type of the returned value) and defines any input Arguments passed to the function. Note that somewhere in the function, a value for GenlFcn must be computed for return to the calling procedure. 



�Function Example:



Here is a Function named CylVol that computes the volume of a cylinder of known height (Height) and radius (Radius).  



Function CylVol(Height As Single, Radius As Single) As Single

Dim Area As Single

Const PI = 3.1415926

Area = PI * Radius ^ 2

CylVol = Area * Height

End Sub



Calling a Function:



To call or use a Function, you equate a variable (of proper type) to the Function, with its arguments.  That is, if the Function GenlFunc is of Type Integer, then use the code segment:



Dim RValue as Integer

	.

	.

RValue = GenlFunc(Arguments)



Example



To call the volume computation function, we could use:



Dim Volume As Single

	.

	.

Volume = CylVol(Height, Radius)



Locating Function Procedures:



Like Sub procedures, Functions can be located in forms or modules.  They are created using exactly the same process described for Sub procedures, the only difference being you use the keyword Function.



And, like Sub procedures, Functions (in modules) can be Public or Private.





�Quick Example:  Temperature Conversion



Open the Temperature Conversion application from last class.  Note in the vsbTemp_Change and vsbTemp_Scroll procedures, there is a lot of repeated code.  We'll replace this code with a Sub procedure that prints the values and a Function procedure that does the temperature conversion.



Add a module to your application.  Create a Function (Public by default) named DegF_To_DegC.



Public Function DegF_To_DegC(DegF As Integer) As Integer

DegF_To_DegC = CInt((DegF - 32) * 5 / 9)

End Function



Go back to your form.  Create a Sub procedure named ShowTemps.  Fill in the code by cutting from an old procedure.  Note this code uses the new Function to convert temperature and prints both values in their respective label boxes.



Private Sub ShowTemps()

lblTempF.Caption = Str(TempF)

TempC = DegF_To_DegC(TempF)

lblTempC.Caption = Str(TempC)

End Sub



No arguments are needed since TempF and TempC are global to the form.  



Rewrite the vsbTemp_Change and vsbTemp_Scroll procedures such that they call the new Sub procedure:



Private Sub vsbTemp_Change()

TempF = vsbTemp.Value

Call ShowTemps

End Sub



Private Sub vsbTemp_Scroll()

Call vsbTemp_Change

End Sub



Note how vsbTemp_Scroll simply calls vsbTemp_Change since they use the same code.  This is an example of calling an event procedure.



Save the application and run it.  Note how much neater and modular the code is.

�Quick Example:  Image Viewer (Optional)



Open the Image Viewer application from last class.  Note the code in the cmdShow_Click and filImage_DblClick events is exactly the same.  Delete the code in the filImage_DblClick procedure and simply have it call the cmdShow_Click procedure.  That is, replace the filImage_DblClick procedure with:



Private Sub filImage_DblClick()

Call cmdShow_Click

End Sub



This is another example of calling an event procedure.  Save your application.





Adding Menus to an Application



As mentioned earlier, it is important that the interface of your application be familar to a seasoned, or not-so-seasoned, Windows user.  One such familiar application component is the Menu bar.  Menus are used to provide a user with choices that control the application.  Menus are easily incorporated into Visual Basic programs using the Menu Editor.



A good way to think about elements of a menu structure is to consider them as a hierarchical list of command buttons that only appear when pulled down from the menu bar.  When you click on a menu item, some action is taken.  Like command buttons, menu items are named, have captions, and have properties.



Example



Here is a typical menu structure:



File	Edit	Format

	New		Cut		Bold

	Open		Copy		Italic

	Save		Paste		Underline

�					Size

	Exit					10

						15

						20



�The underscored characters are access keys, just like those on command buttons.  The level of indentation indicates position of a menu item within the hierarchy.  For example, New is a sub-element of the File menu.  The line under Save in the File menu is a separator bar (separates menu items).  



With this structure, the Menu bar would display:



File	Edit	Format



The sub-menus appear when one of these ‘top’ level menu items is selected.  Note the Size sub-menu under Format has another level of hierarchy.  It is good practice to not use more than two levels in menus.  Each menu element will have a Click event associated with it.



The Menu Editor allows us to define the menu structure, adding access keys and shortcut keys, if desired.  We then add code to the Click events we need to respond to.  The Menu Editor is selected from the Tools menu bar or by clicking the Menu Editor on the toolbar.  This selection can only be made when the form needing the menu is active.  Upon selecting the editor, and entering the example menu structure, the editor window looks like this:



�



Each item in the menu structure requires several entries in this design box.

�The Caption box is where you type the text that appears in the menu bar.  Access keys are defined in the standard way using the ampersand (&).  Separator bars (a horizontal line used to separate menu items) are defined by using a Caption of a single hyphen (-).  When assigning captions and access keys, try to use conform to any established Windows standards.



The Name box is where you enter a control name for each menu item.  This is analogous to the Name property of command buttons and is the name used to set properties and establish the Click event procedure for each menu item.  Each menu item must have a name, even separator bars!  The prefix mnu is used to name menu items.  Sub-menu item names usually refer back to main menu headings.  For example, if the menu item New is under the main heading File menu, use the name mnuFileNew.



The Index box is used for indexing any menu items defined as control arrays.



The Shortcut dropdown box is used to assign shortcut keystrokes to any item in a menu structure.  The shortcut keystroke will appear to the right of the caption for the menu item.  An example of such a keystroke is using Ctrl+X to cut text.



The HelpContextID and NegotiatePosition boxes relate to using on-line help and object linking embedding, and are beyond the scope of this discussion.



Each menu item has four properties associated with it.  These properties can be set at design time using the Menu Editor or at run-time using the standard dot notation.  These properties are:



Checked	Used to indicate whether a toggle option is turned on or off.  If True, a check mark appears next to the menu item.

Enabled	If True, menu item can be selected.  If False, menu item is grayed and cannot be selected.

Visible	Controls whether the menu item appears in the structure.

WindowList	Used with Multiple Document Interface (MDI) - not discussed here.



At the bottom of the Menu Editor form is a list box displaying the hierarchical list of menu items.  Sub-menu items are indented to their level in the hierarchy.  The right and left arrows adjust the levels of menu items, while the up and down arrows move items within the same level.  The Next, Insert, and Delete buttons are used to move the selection down one line, insert a line above the current selection, or delete the current selection, respectively.

�Let’s look at the process of entering the example menu structure.  To do this, we ‘stack’ the three menus on top of each other, that is enter items as a long list.  For each item in this list, we provide a Caption (with access key, if any), a Name (indicative of where it is in the structure), a shortcut key (if any), and provide proper indentation to indication hierarchical location (use the left and right arrows to move in and out).  



After entering this structure, the complete list box at the bottom of the Menu Editor would look like this (notice access keys are indicated with ampersands and shortcut keys are listed at the right, and, the assigned names are shown at the left - these don’t really appear in the Menu Editor list box; they are shown to illustrate one possible naming convention):



�Name

mnuFile	&File

mnuFileNew		&New

mnuFileOpen		&Open

mnuFileSave		&Save

mnuFileBar		-

mnuFileExit		E&xit

mnuEdit	&Edit

mnuEditCut		Cu&t		Ctrl+X

mnuEditCopy		&Copy	Ctrl+C

mnuEditPaste		&Paste	Ctrl+V

mnuFmt	F&ormat

mnuFmtBold		Bold

mnuFmtItalic		Italic

mnuFmtUnderline		Underline

mnuFmtSize		Size

mnuFmtSize10		10

mnuFmtSize15		15

mnuFmtSize20		20





At first, using the Menu Editor may seem a little clunky.  After you’ve done a couple of menu structures, however, its use becomes obvious.  A neat thing is:  after setting up your menu, you can look at it in the Visual Basic design mode and see if it looks like it should.  In the next example, you’ll get practice setting up a similar menu structure.



�Example 5-1



Note Editor



Start a new project.  We will use this application the rest of this class.  We will build a note editor with a menu structure that allows us to control the appearance of the text in the editor box.  Since this is the first time we’ve built menus, I’ll provide the steps involved.



Place a large text box on a form.  Set the properties of the form and text box:



Form1:

	BorderStyle	1-Fixed Single

	Caption	Note Editor

	Name	frmEdit



Text1:

	BorderStyle	1-Fixed Single

	MultiLine	True

	Name	txtEdit

	ScrollBars	2-Vertical

	Text	[Blank]



The form should look something like this when you’re done:



�



�We want to add this menu structure to the Note Editor:



File	Format

	New		Bold

�			Italic

	Exit		Underline

			Size

				Small

				Medium

				Large



Note the identified access keys.  Bring up the Menu Editor and assign the following Captions, Names, and Shortcut Keys to each item.  Make sure each menu item is at is proper location in the hierarchy.



Caption	Name	Shortcut

&File	mnuFile	[None]

&New	mnuFileNew	[None]

-	mnuFileBar	[None]

E&xit	mnuFileExit	[None]

F&ormat	mnuFmt	[None]

& Bold	mnuFmt Bold	Ctrl+B

&Italic	mnuFmtItalic	Ctrl+I

&Underline	mnuFmtUnderline	Ctrl+U

&Size	mnuFmtSize	[None]

&Small	mnuFmtSizeSmall	Ctrl+S

&Medium	mnuFmtSizeMedium	Ctrl+M

&Large	mnuFmtSizeLarge	Ctrl+L



The Small item under the Size sub-menu should also be Checked to indicate the initial font size.  When done, look through your menu structure in design mode to make sure it looks correct.  With a menu, the form will appear like:



�

�Each menu item that performs an action requires code for its Click event.  The only menu items that do not have events are the menu and sub-menu headings, namely File, Format, and Size.  All others need code.  Use the following code for each menu item Click event.  (This may look like a lot of typing, but you should be able to use a lot of cut and paste.)



If mnuFileNew is clicked, the program checks to see if the user really wants a new file and, if so (the default response), clears out the text box:



Private Sub mnuFileNew_Click()

'If user wants new file, clear out text

Dim Response As Integer

Response = MsgBox("Are you sure you want to start a new file?", vbYesNo + vbQuestion, "New File")

If Response = vbYes Then txtEdit.Text = ""

End Sub



If mnuFileExit is clicked, the program checks to see if the user really wants to exit.  If not (the default response), the user is returned to the program:



Private Sub mnuFileExit_Click()

'Make sure user really wants to exit

Dim Response As Integer

Response = MsgBox("Are you sure you want to exit the note editor?", vbYesNo + vbCritical + vbDefaultButton2, "Exit Editor")

If Response = vbNo Then

  Exit Sub

Else

  End

End If

End Sub



If mnuFmtBold is clicked, the program toggles the current bold status:



Private Sub mnuFmtBold_Click()

'Toggle bold font status

mnuFmtBold.Checked = Not (mnuFmtBold.Checked)

txtEdit.FontBold = Not (txtEdit.FontBold)

End Sub



�If mnuFmtItalic is clicked, the program toggles the current italic status:



Private Sub mnuFmtItalic_Click()

'Toggle italic font status

mnuFmtItalic.Checked = Not (mnuFmtItalic.Checked)

txtEdit.FontItalic = Not (txtEdit.FontItalic)

End Sub



If mnuFmtUnderline is clicked, the program toggles the current underline status:



Private Sub mnuFmtUnderline_Click()

'Toggle underline font status

mnuFmtUnderline.Checked = Not (mnuFmtUnderline.Checked)

txtEdit.FontUnderline = Not (txtEdit.FontUnderline)

End Sub



If either of the three size sub-menus is clicked, indicate the appropriate check mark location and change the font size:



Private Sub mnuFmtSizeSmall_Click()

'Set font size to small

mnuFmtSizeSmall.Checked = True

mnuFmtSizeMedium.Checked = False

mnuFmtSizeLarge.Checked = False

txtEdit.FontSize = 8

End Sub



Private Sub mnuFmtSizeMedium_Click()

'Set font size to medium

mnuFmtSizeSmall.Checked = False

mnuFmtSizeMedium.Checked = True

mnuFmtSizeLarge.Checked = False

txtEdit.FontSize = 12

End Sub



Private Sub mnuFmtSizeLarge_Click()

'Set font size to large

mnuFmtSizeSmall.Checked = False

mnuFmtSizeMedium.Checked = False

mnuFmtSizeLarge.Checked = True

txtEdit.FontSize = 18

End Sub

�Save your application.  We will use it again in Class 6 where we’ll learn how to save and open text files created with the Note Editor.  Test out all the options.  Notice how the toggling of the check marks works.  Try the shortcut keys.





Using Pop-Up Menus



Pop-up menus can show up anywhere on a form, usually being activated by a single or double-click of one of the two mouse buttons.  Most Windows applications, and Windows itself, use pop-up menus.  For example, using the right hand mouse button on almost any object in Windows 95 will display a pop-up menu.  In fact, with the introduction of such pop-up menus with Windows 95, the need for adding such menus to Visual Basic applications has been reduced.



Adding pop-up menus to your Visual Basic application is a two step process.  First, you need to create the menu using the Menu Editor (or, you can use any existing menu structure with at least one sub-menu).  If creating a unique pop-up menu (one that normally does not appear on the menu bar), it’s Visible property is set to be False at design time.  Once created, the menu is displayed on a form using the PopupMenu method.



The PopupMenu method syntax is:



ObjectName.PopupMenu MenuName, Flags, X, Y



The ObjectName can be omitted if working with the current form.  The arguments are:



MenuName	Full-name of the pop-up menu to display.

Flags	Specifies location and behavior of menu (optional).

X, Y	(X, Y) coordinate of menu in twips (optional; if either value is omitted, the current mouse coordinate is used).



The Flags setting is the sum of two constants.  The first constant specifies location:



Value	Meaning	Symbolic Constant

0	Left side of menu is at X coordinate	vbPopupMenuLeftAlign

4	Menu is centered at X coordinate	vbPopupMenuCenterAlign

8	Right side of menu is at X coordinate	vbPopupMenuRightAlign



�The second specifies behavior:



Value	Meaning	Symbolic Constant

0	Menu reacts only to left mouse button	vbPopupMenuLeftButton

2	Menu reacts to either mouse button	vbPopupMenuRightButton



You need to decide where to put the code that displays the pop-up menu, that is the PopupMenu method.  Usually, a pop-up menu is displayed in response to a Click event or MouseDown event.  The standard (starting with Windows 95) seems to be leaning toward displaying a pop-up menu in response to a right mouse button click. 



Just like other menus, each item on your pop-up menu will need code for the corresponding Click event.  Many times, though, the code is simply a call to an existing menu item’s Click event.





Assigning Icons to Forms



Notice that whenever you run an application, a small icon appears in the upper left hand corner of the form.  This icon is also used to represent the form when it is minimized at run-time.  The icon seen is the default Visual Basic icon for forms.  Using the Icon property of a form, you can change this displayed icon.



The idea is to assign a unique icon to indicate the form’s function.  To assign an icon, click on the Icon property in the Property Window for the form.  Click on the ellipsis (...) and a window that allows selection of icon files will appear.



The icon file you load must have the .ico filename extension and format.  When you first assign an icon to a form (at design time), it will not appear on the form.  It will only appear after you have run the application once.





Designing Your Own Icon with IconEdit



Visual Basic offers a wealth of icon files from which you could choose an icon to assign to your form(s).  But, it’s also fun to design your own icon to add that personal touch. 



PC Magazine offers a free utility called IconEdit that allows you to design and save icons.  Included with these notes is this program and other files (directory IconEdit).  To install these files on your machine, copy the folder to your hard drive.

�To run IconEdit, click Start on the Windows 95 task bar, then click Run.  Find the IconEdit.exe program (use Browse mode).  You can also establish an shortcut to start IconEdit from your desktop, if desired.  The following Editor window will appear:



�



The basic idea of IconEdit is to draw an icon in the 32 x 32 grid displayed.  You can draw single points, lines, open rectangles and ovals, and filled rectangles and ovals.  Various colors are available.  Once completed, the icon file can be saved for attaching to a form.



Another fun thing to do with IconEdit is to load in Visual Basic icon files and see how much artistic talent really goes into creating an icon.  



We won’t go into a lot of detail on using the IconEdit program here - I just want you to know it exists and can be used to create and save icon files.  Its use is fairly intuitive.  Consult the on-line help of the program for details.  And, there is a .txt file included that is very helpful.





�Creating Visual Basic Executable Files



Up to now, to run any of the applications created, we needed Visual Basic.  The goal of creating an application is to let others (without Visual Basic) use it.  This is accomplished by creating an executable version of the application.



Before creating an executable, you should make sure your program is free of bugs and operating as desired.  Save all forms, modules, and project files.  Any later changes to the application will require re-creating the executable file.



The executable file will have the extension .exe.  To create an exe file for your application, select Make [Project name] exe from Visual Basic’s File menu.  This will display the Make EXE File dialog box, where you name the exe file.  To open the Options box, click that button.  The EXE Options dialog box will appear:



�



We’ll only concern ourselves with two pieces of information in this box:  Title and Icon.  The Title is the name you want to give your application.  It does not have to be the same as the Project name.  The Icon is selected from icons assigned to form(s) in your application.  The selected icon is used to identify the application everywhere it is needed in Windows 95.



Once you have selected these options, return to the Make EXE File dialog box, select a directory (best to use the same directory your application files are in) and name for your executable file.  Click OK and the exe file is created.  

�Use Windows Explorer to confirm creation of the file.  And, while there, double-click the executable file name and the program will run!

�Example 5-2



Note Editor - Building an Executable and Attaching an Icon



Open your Note Editor project.  Attach an icon to the form by setting the Icon property.  If you want to, open up the Image Viewer project from last class to take a look at icon files.  The icon I used is note.ico



Create an executable version of your Note Editor.  Confirm creation of the exe file and run it under the Windows Explorer.



Something you might want to try with your application is create a Windows 95 shortcut to run your program, that is, display a clickable icon.  To get started, click the Start button on the taskbar, then Settings, then Taskbar.  Here, you can add programs to those that show up when you select Start.  The process is straightforward.  Find your application, specify the folder you want the shortcut to appear in, and name your application.  When done, the icon will appear in the specified location.

�Using the Visual Basic Package & Deployment Wizard



We were able to run the Note Editor executable file because Visual Basic is installed on our system.  If you gave someone a copy of your exe file and they tried to run it, it wouldn’t work (unless they have Visual Basic installed also).  The reason it wouldn’t run is that the executable file also needs some ancillary files (primarily, so-called dynamic link libraries) to run properly.  These libraries provide most of the code associated with keeping things on a form working properly.



So to allow others to run your application, you need to give them the executable file (exe) and at least two dynamic link libraries.  Unfortunately, these dynamic link libraries take up over 1 Meg of disk space, so it’s hard to move those around on a floppy disk.



Visual Basic solves this ‘distribution problem’ by providing a very powerful tool called the Visual Basic Package & Deployment Wizard.  This wizard is installed along with Visual Basic.



The Package & Deployment Wizard prepares your application for distribution.  It helps you determine which files to distribute, creates a Setup program (written in Visual Basic) that works like all other Windows Setup programs (setup.exe), compresses all required files to save disk space, and writes the compressed files to the distribution media of choice, usually floppy disk(s).  



To start the Package & Deployment Wizard, click the Start button in Windows, then find the Visual Basic  program folder - click on Visual Basic Tools, then choose Package & Deployment Wizard  The setup follows several steps.  The directions provided with each step pertain to the simple applications we develop in class.  For more complicated examples, you need to modify the directions, especially regarding what files you need to distribute with your application.



�Step 1.	Initial Information.  Enter the path and file name for your project file (.vbp).  Click the ellipsis (...) to browse vbp files.  If an executable (.exe) file does not exist, one will be created.  Click the ‘Package’ button to continue.  If you have previously saved a setup package for the selected project, it will load the package file created during that session.



�



Step 2.	Package Type.  Choose the Standard Setup Package (we want a standard setup program).  Click Next to continue.



Step 3.	Package Folder.  Select a directory where you want the application distribution package to be saved.  Click Next to continue.  Click Back to return to the previous step.



Step 4.	Included Files.  The Package & Deployment Wizard will list all files it believes are required for your application to function properly.  If your application requires any files not found by the wizard (for example, external data files you have created), you would add them to the setup list here (click Add).  To continue, click Next.  Click Back to return to the previous step.



Step 5.	Cab Options.  Distribution files are called cab files (have a cab extension).  You can choose a Single cab file written to your hard drive (if you use CD ROM distribution), or Multiple cab files (to allow distribution on floppy disks).  If you choose, Multiple, you also specify the capacity of the disks you will use to write your distribution file(s).  Make your choice.  Click Next to Continue.  Click Back to return to the previous step. 



Step 6.	Installation Title.  Enter a title you want your application to have.  Click Next to Continue.  Click Back to return to previous step.



Step 7.	Start Menu Items.  This step determines where your installed application will be located on the user’s Start menu.  We will use the default choice.  Click Next to Continue.  Click Back to return to previous step.



Step 8.	Install Locations.  The wizard gives you an opportunity to change the locations of installed files.  Click Next to Continue.  Click Back to return to previous step.



Step 9.	Shared Files.  Some files in your application may be shared by other applications.  Shared files will not be removed if the application is uninstalled.  Decide if you have shared files.  Click Next to Continue.  Click Back to return to previous step.



Step 10.	Finished!  Provide a name for saving the script for this wizard session (a file that saves answers to all the questions you just answered).  Click Finish to Continue.  Click Back to return to previous step.  The wizard will create and write the cab files and tell you where they are saved.  Click Close.  You will be returned to the Package & Deployment Wizard opening window.  Click Close.



Step 11.	Write Distribution Media.  This is not a step in the wizard, but one you must take.  The cab files (distribution files) the wizard wrote must now be copied to your distribution media.  If you wrote a single cab file (for CD ROM), copy that file, the setup.exe file (the setup application), and the setup.lst file to your CD ROM).  If you wrote multiple files (for floppy disks), copy the setup.exe, setup.lst, and first cab file (1 at end of file name) to floppy number 1.  Copy the second cab file (2 at end of file name) to floppy number 2.  Copy all subsequent cab files to as many floppies as needed.  Properly label all floppies.



To install the application using the distribution CD ROM or floppy disk(s), a user simply puts CD ROM or floppy number 1 in a drive.  Then, through the Windows Explorer, run the setup.exe program.  The user will be taken through the installation procedure step-by-step.  The procedure is nearly identical to the installation process for all Microsoft products.



The Package & Deployment Wizard is a very powerful tool.  We’ve only looked at using it for simple applications.  As your programming skills begin to include database access and other advanced topics, you will need to become familiar with other files that should be distributed with your applications.



�Example 5-3



Note Editor - Creating a Distribution Disk



Open your Note Editor project again.  Create a distribution disk using the Package & Deployment Wizard.



Try installing the application on your computer.  Better yet, take the disk to another Windows 95/98/NT-based machine, preferably without Visual Basic installed.  Install the application using the distribution disk and test its operation.





�

�Exercise 5



US Capitals Quiz



Develop an application that quizzes a user on states and capitals in the United States.  Use a menu structure that allows the user to decide whether they want to name states or capitals and whether they want mulitple choice or type-in answers.  Throughly test your application.  Design an icon for your program using IconEdit or some other program.  Create an executable file.  Create a distribution disk using the Application Setup Wizard.  Give someone your application disk and have them install it on their computer and try out your nifty little program.





My Solution:



Form:



�

�

���������������������������



�Properties:



Form frmCapitals:

	BorderStyle = 1 - Fixed Single

	Caption = US Capitals



CommandButton cmdNext:

	Caption = &Next Question

	Enabled = False



CommandButton cmdExit:

	Caption = E&xit



TextBox txtAnswer:

	FontName = MS Sans Serif

	FontSize = 13.2

	Visible = False



Label lblComment:

	Alignment = 2 - Center

	BackColor = &H00C00000& (Blue)

	BorderStyle = 1 - Fixed Single

	FontName = MS Sans Serif

	FontSize = 13.2

	FontItalic = True

	ForeColor = &H0000FFFF& (Yellow)



Label lblScore:

	Alignment = 2 - Center

	AutoSize = True

	BackColor = &H0000FFFF& (Yellow)

	BorderStyle = 1 - Fixed Single

	Caption = 0%

	FontName = MS Sans Serif

	FontSize = 15.6

	FontBold = True



Label lblAnswer (control array):

	Alignment = 2 - Center

	BackColor = &H00FFFFFF& (White)

	BorderStyle = 1 - Fixed Single

	FontName = MS Sans Serif

	FontSize = 13.2

	Index = 0, 1, 2, 3

�Label lblHeadAnswer:

	Caption = Capital:

	FontName = MS Sans Serif

	FontSize = 13.2

	FontBold = True 



Label lblHeadGiven:

	Caption = State:

	FontName = MS Sans Serif

	FontSize = 13.2

	FontBold = True



Menu mnuFile:

	Caption = &File



Menu mnuFileNew:

	Caption = &New



Menu mnuFileBar:

	Caption = -



Menu mnuFileExit:

	Caption = E&xit



Menu mnuOptions:

	Caption = &Options



Menu mnuOptionsCapitals:

	Caption = Name &Capitals

	Checked = True



Menu mnuOptionsState:

	Caption = Name &State



Menu mnuOptionsBar:

	Caption = -



Menu mnuOptionsMC:

	Caption = &Multiple Choice Answers

	Checked = True



Menu mnuOptionsType:

	Caption = &Type In Answers

�Code:



General Declarations:



Option Explicit

Dim CorrectAnswer As Integer

Dim NumAns As Integer, NumCorrect As Integer

Dim Wsound(26) As Integer

Dim State(50) As String, Capital(50) As String





SoundEx General Function (this is a neat little function to check if spelling of two words is similar):



Private Function SoundEx(W As String, Wsound() As Integer) As String

‘Generates Soundex code for W

‘Allows answers whose spelling is close, but not exact

Dim Wtemp As String, S As String

Dim L As Integer, I As Integer

Dim Wprev As Integer, Wsnd As Integer, Cindex As Integer

Wtemp = UCase(W)

L = Len(W)

If L <> 0 Then

  S = Left(Wtemp, 1)

  Wprev = 0

  If L > 1 Then

    For I = 2 To L

      Cindex = Asc(Mid(Wtemp, I, 1)) - 64

      If Cindex >= 1 And Cindex <= 26 Then

        Wsnd = Wsound(Cindex) + 48

        If Wsnd <> 48 And Wsnd <> Wprev Then S = S + Chr(Wsnd)

        Wprev = Wsnd

      End If

    Next I

  End If

Else

  S = ""

End If

SoundEx = S

End Function

�Update_Score General Procedure:



Private Sub Update_Score(Iscorrect As Integer)

Dim I As Integer

'Check if answer is correct

cmdNext.Enabled = True

cmdNext.SetFocus

If Iscorrect = 1 Then

  NumCorrect = NumCorrect + 1

  lblComment.Caption = "Correct!"

Else

  lblComment.Caption = "Sorry ..."

End If

'Display correct answer and update score

If mnuOptionsMC.Checked = True Then

  For I = 0 To 3

    If mnuOptionsCapitals.Checked = True Then

      If lblAnswer(I).Caption <> Capital(CorrectAnswer) Then

        lblAnswer(I).Caption = ""

      End If

    Else

      If lblAnswer(I).Caption <> State(CorrectAnswer) Then

        lblAnswer(I).Caption = ""

      End If

    End If

  Next I

Else

  If mnuOptionsCapitals.Checked = True Then

    txtAnswer.Text = Capital(CorrectAnswer)

  Else

    txtAnswer.Text = State(CorrectAnswer)

  End If

End If

lblScore.Caption = Format(NumCorrect / NumAns, "##0%")

End Sub





cmdExit Click Event:



Private Sub cmdExit_Click()

'Exit program

Call mnuFileExit_Click

End Sub





�cmdNext Click Event:



Private Sub cmdNext_Click()

'Generate the next question

cmdNext.Enabled = False

Call Next_Question(CorrectAnswer)

End Sub





Form Activate Event:



Private Sub Form_Activate()

Call mnufilenew_click

End Sub





Form Load Event:



Private Sub Form_Load()

Randomize Timer

'Load soundex function array

Wsound(1) = 0: Wsound(2) = 1: Wsound(3) = 2: Wsound(4) = 3

Wsound(5) = 0: Wsound(6) = 1: Wsound(7) = 2: Wsound(8) = 0

Wsound(9) = 0: Wsound(10) = 2: Wsound(11) = 2: Wsound(12) = 4

Wsound(13) = 5: Wsound(14) = 5: Wsound(15) = 0: Wsound(16) = 1

Wsound(17) = 2: Wsound(18) = 6: Wsound(19) = 2: Wsound(20) = 3

Wsound(21) = 0: Wsound(22) = 1: Wsound(23) = 0: Wsound(24) = 2

Wsound(25) = 0: Wsound(26) = 2

'Load state/capital arrays

State(1) = "Alabama": Capital(1) = "Montgomery"

State(2) = "Alaska": Capital(2) = "Juneau"

State(3) = "Arizona": Capital(3) = "Phoenix"

State(4) = "Arkansas": Capital(4) = "Little Rock"

State(5) = "California": Capital(5) = "Sacramento"

State(6) = "Colorado": Capital(6) = "Denver"

State(7) = "Connecticut": Capital(7) = "Hartford"

State(8) = "Delaware": Capital(8) = "Dover"

State(9) = "Florida": Capital(9) = "Tallahassee"

State(10) = "Georgia": Capital(10) = "Atlanta"

State(11) = "Hawaii": Capital(11) = "Honolulu"

State(12) = "Idaho": Capital(12) = "Boise"

State(13) = "Illinois": Capital(13) = "Springfield"

State(14) = "Indiana": Capital(14) = "Indianapolis"

State(15) = "Iowa": Capital(15) = "Des Moines"

State(16) = "Kansas": Capital(16) = "Topeka"

State(17) = "Kentucky": Capital(17) = "Frankfort"

State(18) = "Louisiana": Capital(18) = "Baton Rouge"

State(19) = "Maine": Capital(19) = "Augusta"

State(20) = "Maryland": Capital(20) = "Annapolis"

State(21) = "Massachusetts": Capital(21) = "Boston"

State(22) = "Michigan": Capital(22) = "Lansing"

State(23) = "Minnesota": Capital(23) = "Saint Paul"

State(24) = "Mississippi": Capital(24) = "Jackson"

State(25) = "Missouri": Capital(25) = "Jefferson City"

State(26) = "Montana": Capital(26) = "Helena"

State(27) = "Nebraska": Capital(27) = "Lincoln"

State(28) = "Nevada": Capital(28) = "Carson City"

State(29) = "New Hampshire": Capital(29) = "Concord"

State(30) = "New Jersey": Capital(30) = "Trenton"

State(31) = "New Mexico": Capital(31) = "Santa Fe"

State(32) = "New York": Capital(32) = "Albany"

State(33) = "North Carolina": Capital(33) = "Raleigh"

State(34) = "North Dakota": Capital(34) = "Bismarck"

State(35) = "Ohio": Capital(35) = "Columbus"

State(36) = "Oklahoma": Capital(36) = "Oklahoma City"

State(37) = "Oregon": Capital(37) = "Salem"

State(38) = "Pennsylvania": Capital(38) = "Harrisburg"

State(39) = "Rhode Island": Capital(39) = "Providence"

State(40) = "South Carolina": Capital(40) = "Columbia"

State(41) = "South Dakota": Capital(41) = "Pierre"

State(42) = "Tennessee": Capital(42) = "Nashville"

State(43) = "Texas": Capital(43) = "Austin"

State(44) = "Utah": Capital(44) = "Salt Lake City"

State(45) = "Vermont": Capital(45) = "Montpelier"

State(46) = "Virginia": Capital(46) = "Richmond"

State(47) = "Washington": Capital(47) = "Olympia"

State(48) = "West Virginia": Capital(48) = "Charleston"

State(49) = "Wisconsin": Capital(49) = "Madison"

State(50) = "Wyoming": Capital(50) = "Cheyenne"

End Sub





�lblAnswer Click Event:



Private Sub lblAnswer_Click(Index As Integer)

'Check multiple choice answers

Dim Iscorrect As Integer

'If already answered, exit

If cmdNext.Enabled = True Then Exit Sub

Iscorrect = 0

If mnuOptionsCapitals.Checked = True Then

  If lblAnswer(Index).Caption = Capital(CorrectAnswer) Then Iscorrect = 1

Else

  If lblAnswer(Index).Caption = State(CorrectAnswer) Then Iscorrect = 1

End If

Call Update_Score(Iscorrect)

End Sub





mnuFileExit Click Event:



Private Sub mnuFileExit_Click()

'End the application

End

End Sub





mnuFileNew Click Event:



Private Sub mnufilenew_click()

'Reset the score and start again

NumAns = 0

NumCorrect = 0

lblScore.Caption = "0%"

lblComment.Caption = ""

cmdNext.Enabled = False

Call Next_Question(CorrectAnswer)

End Sub





�mnuOptionsCapitals Click Event:



Private Sub mnuOptionsCapitals_Click()

'Set up for providing capital, given state

mnuOptionsState.Checked = False

mnuOptionsCapitals.Checked = True

lblHeadGiven.Caption = "State:"

lblHeadAnswer.Caption = "Capital:"

Call mnufilenew_click

End Sub





mnuOptionsMC Click Event:



Private Sub mnuOptionsMC_Click()

'Set up for multiple choice answers

Dim I As Integer

mnuOptionsMC.Checked = True

mnuOptionsType.Checked = False

For I = 0 To 3

  lblAnswer(I).Visible = True

Next I

txtAnswer.Visible = False

Call mnufilenew_click

End Sub





mnuOptionsState Click Event:



Private Sub mnuOptionsState_Click()

'Set up for providing state, given capital

mnuOptionsState.Checked = True

mnuOptionsCapitals.Checked = False

lblHeadGiven.Caption = "Capital:"

lblHeadAnswer.Caption = "State:"

Call mnufilenew_click

End Sub





�mnuOptionsType Click Event:



Private Sub mnuOptionsType_Click()

'Set up for type in answers

Dim I As Integer

mnuOptionsMC.Checked = False

mnuOptionsType.Checked = True

For I = 0 To 3

  lblAnswer(I).Visible = False

Next I

txtAnswer.Visible = True

Call mnufilenew_click

End Sub





Next_Question General Procedure:



Private Sub Next_Question(Answer As Integer)

Dim VUsed(50) As Integer, I As Integer, J As Integer

Dim Index(3)

lblComment.Caption = ""

NumAns = NumAns + 1

'Generate the next question based on selected options

Answer = Int(Rnd * 50) + 1

If mnuOptionsCapitals.Checked = True Then

  lblGiven.Caption = State(Answer)

Else

  lblGiven.Caption = Capital(Answer)

End If

If mnuOptionsMC.Checked = True Then

'Multiple choice answers

'Vused array is used to see which states have

'been selected as possible answers

  For I = 1 To 50

    VUsed(I) = 0

  Next I

'Pick four different state indices (J) at random

'These are used to set up multiple choice answers

'Stored in the Index array

  I = 0

  Do

    Do

      J = Int(Rnd * 50) + 1

    Loop Until VUsed(J) = 0 And J <> Answer

    VUsed(J) = 1

    Index(I) = J

    I = I + 1

  Loop Until I = 4

'Now replace one index (at random) with correct answer

  Index(Int(Rnd * 4)) = Answer

'Display multiple choice answers in label boxes

  For I = 0 To 3

    If mnuOptionsCapitals.Checked = True Then

      lblAnswer(I).Caption = Capital(Index(I))

    Else

      lblAnswer(I).Caption = State(Index(I))

    End If

  Next I

Else

'Type-in answers

  txtAnswer.Locked = False

  txtAnswer.Text = ""

  txtAnswer.SetFocus

End If

End Sub





�txtAnswer KeyPress Event:



Private Sub txtAnswer_KeyPress(KeyAscii As Integer)

'Check type in answer'

Dim Iscorrect As Integer

Dim YourAnswer As String, TheAnswer As String

'Exit if already answered

If cmdNext.Enabled = True Then Exit Sub

If (KeyAscii >= vbKeyA And KeyAscii <= vbKeyZ) _

Or (KeyAscii >= vbKeyA + 32 And KeyAscii <= vbKeyZ + 32) _

Or KeyAscii = vbKeySpace Or KeyAscii = vbKeyBack Or KeyAscii = vbKeyReturn Then

'Acceptable keystroke

  If KeyAscii <> vbKeyReturn Then Exit Sub

'Lock text box once answer entered

  txtAnswer.Locked = True

  Iscorrect = 0

'Convert response and correct answers to all upper

'case for typing problems

  YourAnswer = UCase(txtAnswer.Text)

  If mnuOptionsCapitals.Checked = True Then

    TheAnswer = UCase(Capital(CorrectAnswer))

  Else

    TheAnswer = UCase(State(CorrectAnswer))

  End If

'Check for both exact and approximate spellings

  If YourAnswer = TheAnswer Or _

  SoundEx(YourAnswer, Wsound()) = SoundEx(TheAnswer, Wsound()) Then Iscorrect = 1

  Call Update_Score(Iscorrect)

Else

'Unacceptable keystroke

  KeyAscii = 0

End If

End Sub

